

```

POUR courbe2 :n
FIXEHASARD -280 280
EFFACEECRAN EFFACETEXTE CACHETORTUE
PARTAGE [ "x "y ]
DONNE "x [] DONNE "y []
poly :n
DONNE "h 0,05
REPETE 30 [ // 30 itérations. //
EFFACETEXTE
COULEURTORTUE ITEM :couleurs 1+ RESTE BOUCLE 15
DONNE "xxm [] DONNE "yym []
//On calcule les nouvelles coordonnées des points extrêmes des listes :x et :y pour faciliter les
calculs. On peut à présent tronquer ces deux listes au fur et à mesure des itérations, d'où
accélération du tracé. //
DONNE "xd ( DERNIER :x ) + ( :h * ( ( PREMIER :x ) - ( DERNIER :x ) ) )
DONNE "yd ( DERNIER :y ) + ( :h * ( ( PREMIER :y ) - ( DERNIER :y ) ) )
DONNE "xxm PHRASE :xxm ( PREMIER :x ) + ( :h * ( ( ITEM :x 2 ) - ( PREMIER :x ) ) )
DONNE "yym PHRASE :yym ( PREMIER :y ) + ( :h * ( ( ITEM :y 2 ) - ( PREMIER :y ) ) )
SEGMENT [ ITEM :x 2 ITEM :y 2 ] [ DERNIER :xxm DERNIER :yym ]
SEGMENT [ :xd :yd ] [ DERNIER :xxm DERNIER :yym ]
DONNE "x SAUFPREMIER :x DONNE "y SAUFPREMIER :y
REPETE ( :n - 2 ) [ // la courbe précédente poursuit la suivante.//
DONNE "xxm PHRASE :xxm ( PREMIER :x ) + ( :h * ( ( ITEM :x 2 ) - ( PREMIER :x ) ) )
DONNE "yym PHRASE :yym ( PREMIER :y ) + ( :h * ( ( ITEM :y 2 ) - ( PREMIER :y ) ) )
SEGMENT [ ITEM :x 2 ITEM :y 2 ] [ DERNIER :xxm DERNIER :yym ] // on trace ici. //
DONNE "x SAUFPREMIER :x DONNE "y SAUFPREMIER :y
]
DONNE "xxm PHRASE :xxm :xd DONNE "yym PHRASE :yym :yd
DONNE "x [] DONNE "y []
REPETE :n [ // on actualise la liste des nouvelles coordonnées.//
DONNE "x PHRASE :x PREMIER :xxm
DONNE "y PHRASE :y PREMIER :yym
DONNE "xxm SAUFPREMIER :xxm
DONNE "yym SAUFPREMIER :yym
]
]
FIN

POUR poly :n
ECRISLIGNE [ Calcul des points initiaux. ]
REPETE :n [
FIXEHASARD -260 260
DONNE "x PHRASE :x HASARD
DONNE "y PHRASE :y HASARD
FIXEHASARD 0 0,005 ATTENDS HASARD
]
FIN

```

```

donne "couleurs [ blanche verte jaune rouge orange tilleul fuchsia sarcelle marron rose violette argent grise noire bleue
olive ]
courbe2 6

```